10a. Substitution rules

See §3.3.2

The substitution rules work with **tht[m]** as function on N. It is $\vartheta_m = \Theta_{l,c}(h_{l,m})$ in the text.

The sign off **ell** is denoted by **eps**.

dtt[eps] ($\delta_{\pm\varepsilon}$ in the text) is equal to 1 if eps=1 and equal to 0 if eps=-1

In[ ]:=
```
Clear[subnab, tht, t, eps, dtt]
(*eps denotes the sign of ell *)fctlist = Union[fctlist, {t, tht}];
dtt[e_] := (1 + e) / 2
subnab[xx_] := Module[{x}, x = xx /. subnab0 // Expand;
  (x /. subnab1 // Simplify) /. subnab2 // Simplify]
subnab0 = {Rna[HHr, ff_] :> t D[ff, t],
   Rna[XX0, tht[m_]] → t ^ 2 Pi I ell tht[m],
   Rna[XX1, tht[m_]] :>
    - I eps t Sqrt[2 Pi Abs[ell]] (Sqrt[m] tht[m - 1] + Sqrt[m + 1] tht[m + 1]),
   Rna[XX2, tht[m_]] :> t Sqrt[2 Pi Abs[ell]] (Sqrt[m] tht[m - 1] - Sqrt[m + 1] tht[m + 1]),
   Rna[XX0, ff_ /; FreeQ[ff, tht]] :> 0, Rna[XX1, ff_ /; FreeQ[ff, tht]] :> 0,
   Rna[XX2, ff_ /; FreeQ[ff, tht]] :> 0, Rna[HHr, ff_ /; FreeQ[ff, t]] :> 0};
subnab1 = {tht[- 1] → 0, eps ^ 2 → 1};
subnab2 = {};(* to be chosen later *)
```

The relations for the basis elements on **n** are taken from Table 2.4. We check the relations mentioned in §3.3.2:

In[ ]:=
```
Rna[eps XX1 + I XX2, tht[m]] // subnab
Rna[eps XX1 - I XX2, tht[m]] // subnab
```

Out[ ]= $-2 i \sqrt{1+m} \sqrt{2\pi}\ t \sqrt{\text{Abs[ell]}}\ \text{tht}[1+m]$

Out[ ]= $-2 i \sqrt{m} \sqrt{2\pi}\ t \sqrt{\text{Abs[ell]}}\ \text{tht}[-1+m]$

In[ ]:=
```
Clear[F, h, p, r, q]
F = tht[m] × f[t] × Phi[h, p, r, q]
```

Out[ ]= f[t] × Phi[h, p, r, q] × tht[m]

```
In[ ]:= r31 = R[Z31, F] // subnab
Out[ ]= 1/(8 × (1 + p)) (-2 i Sqrt[2 π] t Sqrt[Abs[ell]] f[t]
    ((p - q) Phi[3 + h, -1 + p, -1 + r, 1 + q] + (-2 - p + r) Phi[3 + h, 1 + p, -1 + r, 1 + q])
    ((1 + eps) Sqrt[m] tht[-1 + m] + (-1 + eps) Sqrt[1 + m] tht[1 + m]) +
   tht[m] × (f[t] (-((p - q) (4 - h + 2 p + r - 4 ell π t^2) Phi[3 + h, -1 + p, 1 + r, 1 + q]) +
         (2 + p + r) (h + 2 p - r + 4 ell π t^2) Phi[3 + h, 1 + p, 1 + r, 1 + q]) +
      2 t ((p - q) Phi[3 + h, -1 + p, 1 + r, 1 + q] + (2 + p + r) Phi[3 + h, 1 + p, 1 + r, 1 + q]) f′[t]))
```

```
In[ ]:= r23 = R[Z23, F] // subnab
Out[ ]= 1/(8 × (1 + p)) (-2 i Sqrt[2 π] t Sqrt[Abs[ell]] f[t]
    ((p - q) Phi[-3 + h, -1 + p, 1 + r, 1 + q] + (2 + p + r) Phi[-3 + h, 1 + p, 1 + r, 1 + q])
    ((-1 + eps) Sqrt[m] tht[-1 + m] + (1 + eps) Sqrt[1 + m] tht[1 + m]) +
   tht[m] × (f[t] ((p - q) (4 + h + 2 p - r + 4 ell π t^2) Phi[-3 + h, -1 + p, -1 + r, 1 + q] +
         (2 + p - r) (-h + 2 p + r - 4 ell π t^2) Phi[-3 + h, 1 + p, -1 + r, 1 + q]) +
      2 t ((-p + q) Phi[-3 + h, -1 + p, -1 + r, 1 + q] + (2 + p - r) Phi[-3 + h, 1 + p, -1 + r, 1 + q]) f′[t]))
```

```
In[ ]:= r13 = R[Z13, F] // subnab
Out[ ]= 1/(8 × (1 + p)) (2 i Sqrt[2 π] t Sqrt[Abs[ell]] f[t]
    ((p + q) Phi[-3 + h, -1 + p, 1 + r, -1 + q] - (2 + p + r) Phi[-3 + h, 1 + p, 1 + r, -1 + q])
    ((-1 + eps) Sqrt[m] tht[-1 + m] + (1 + eps) Sqrt[1 + m] tht[1 + m]) -
   tht[m] × (f[t] ((p + q) (4 + h + 2 p - r + 4 ell π t^2) Phi[-3 + h, -1 + p, -1 + r, -1 + q] +
         (2 + p - r) (h - 2 p - r + 4 ell π t^2) Phi[-3 + h, 1 + p, -1 + r, -1 + q]) -
      2 t ((p + q) Phi[-3 + h, -1 + p, -1 + r, -1 + q] + (2 + p - r) Phi[-3 + h, 1 + p, -1 + r, -1 + q]) f′[t]))
```

```
In[ ]:= r32 = R[Z32, F] // subnab
Out[ ]= -1/(8 × (1 + p)) (2 i Sqrt[2 π] t Sqrt[Abs[ell]] f[t]
    ((p + q) Phi[3 + h, -1 + p, -1 + r, -1 + q] + (2 + p - r) Phi[3 + h, 1 + p, -1 + r, -1 + q])
    ((1 + eps) Sqrt[m] tht[-1 + m] + (-1 + eps) Sqrt[1 + m] tht[1 + m]) +
   tht[m] × (f[t] ((p + q) (4 - h + 2 p + r - 4 ell π t^2) Phi[3 + h, -1 + p, 1 + r, -1 + q] +
         (2 + p + r) (h + 2 p - r + 4 ell π t^2) Phi[3 + h, 1 + p, 1 + r, -1 + q]) +
      2 t (-((p + q) Phi[3 + h, -1 + p, 1 + r, -1 + q]) + (2 + p + r) Phi[3 + h, 1 + p, 1 + r, -1 + q]) f′[t]))
```

The dependence on **eps** can be written in a shorter way.

In[ ]:=
```
subnab2 := {(1 + eps) Sqrt[m] tht[-1 + m] + (-1 + eps) Sqrt[1 + m] tht[1 + m] →
    2 eps Sqrt[m + dtt[-eps]] tht[m - eps],
   (-1 + eps) Sqrt[m] tht[-1 + m] + (1 + eps) Sqrt[1 + m] tht[1 + m] →
    2 eps Sqrt[m + dtt[eps]] tht[m + eps]};
```

This changes the behavior of subnab

For the checks in of the differentiation, and of the shift operators in Table 3.9 we want to work with **dtt** as a formal quantity. Hence here and in the next subsection dtt has to be cleared. After reinitialization the original meaning is restored.

In[ ]:=
```
Clear[dtt]
```

In[ ]:=
```
(R[Z31, F] // subnab) == r31 // Simplify
% //. subnab2 // FullSimplify
```

Out[ ]=

$$\frac{1}{1+p} t \sqrt{\text{Abs[ell]}}\, f[t] \left((p-q)\, \text{Phi}[3+h, -1+p, -1+r, 1+q] + (-2-p+r)\, \text{Phi}[3+h, 1+p, -1+r, 1+q]\right) \left((1+\text{eps}) \sqrt{m}\, \text{tht}[-1+m] + (-1+\text{eps}) \sqrt{1+m}\, \text{tht}[1+m] - 2\, \text{eps} \sqrt{m+\text{dtt}[-\text{eps}]}\, \text{tht}[-\text{eps}+m]\right) == 0$$

Out[ ]= True

In[ ]:=
```
(R[Z23, F] // subnab) == r23 // Simplify
% //. subnab2
```

Out[ ]=

$$\frac{1}{1+p} t \sqrt{\text{Abs[ell]}}\, f[t] \left((p-q)\, \text{Phi}[-3+h, -1+p, 1+r, 1+q] + (2+p+r)\, \text{Phi}[-3+h, 1+p, 1+r, 1+q]\right) \left((-1+\text{eps}) \sqrt{m}\, \text{tht}[-1+m] + (1+\text{eps}) \sqrt{1+m}\, \text{tht}[1+m] - 2\, \text{eps} \sqrt{m+\text{dtt}[\text{eps}]}\, \text{tht}[\text{eps}+m]\right) == 0$$

Out[ ]= True

```
In[ ]:= (R[Z13, F] // subnab) == r13 // Simplify
        % //. subnab2
        (R[Z32, F] // subnab) == r32 // Simplify
        % //. subnab2
```

Out[ ]=
$$\frac{1}{1+p} t \sqrt{\text{Abs[ell]}}\ \text{f[t]} ((p+q)\ \text{Phi}[-3+h, -1+p, 1+r, -1+q] - (2+p+r)\ \text{Phi}[-3+h, 1+p, 1+r, -1+q]) \left((-1+\text{eps}) \sqrt{m}\ \text{tht}[-1+m] + (1+\text{eps}) \sqrt{1+m}\ \text{tht}[1+m] - 2\ \text{eps} \sqrt{m+\text{dtt[eps]}}\ \text{tht}[\text{eps}+m]\right) == 0$$

Out[ ]= True

Out[ ]=
$$\frac{1}{1+p} t \sqrt{\text{Abs[ell]}}\ \text{f[t]} ((p+q)\ \text{Phi}[3+h, -1+p, -1+r, -1+q] + (2+p-r)\ \text{Phi}[3+h, 1+p, -1+r, -1+q]) \left((1+\text{eps}) \sqrt{m}\ \text{tht}[-1+m] + (-1+\text{eps}) \sqrt{1+m}\ \text{tht}[1+m] - 2\ \text{eps} \sqrt{m+\text{dtt}[-\text{eps}]}\ \text{tht}[-\text{eps}+m]\right) == 0$$

Out[ ]= True

In this way we can handle both cases of the sign of ell in the same formula.